

국가승인통계
(승인번호 제117002호)

최초 공개 : 2018년 12월
최근 업데이트 및 수정 : 2021년 2월

국민건강영양조사 가속도계 원시자료 이용지침서



질병관리청

이용자를 위하여

1

국민건강영양조사는 국민건강증진법 제16조에 근거하여 시행하는 국민의 건강행태, 만성질환 유병현황, 식품 및 영양섭취실태에 관한 법정조사이며, 통계법 제17조에 근거한 정부지정통계(승인번호 제117002호)입니다.

2

국민건강영양조사는 제1기(1998)부터 제3기(2005)까지 3년 주기로 실시하였고, 이후 연중 조사체제로 개편되어 제4기(2007-2009)부터 현재까지 매년 실시하고 있습니다. 본 조사는 질병관리청 연구윤리심의위원회 승인을 받아 수행하였고, 조사익년도 12월까지 보도자료 배포, 통계집 발간, 원시자료 공개를 통해 조사결과를 공표하고 있습니다.

※국민건강영양조사 누리집(<http://knhanes.kdca.go.kr/>) 자료실을 방문하시면 보도자료, 통계집, 조사표, 조사지침서, 질관리 보고서, 표본설계 보고서, 기타 연구용역 결과보고서 등의 자료를 다운로드 받으실 수 있습니다.

〈질병관리청 연구윤리심의위원회 승인번호〉

	1차년도	2차년도	3차년도
제4기 (2007-2009)	2007-02CON-04-P	2008-04EXP-01-C	2009-01CON-03-2C
제5기 (2010-2012)	2010-02CON-21-C	2011-02CON-06-C	2012-01EXP-01-2C
제6기 (2013-2015)	2013-07CON-03-4C	2013-12EXP-03-5C	- (*)
제7기 (2016-2018)	-(*)	-(*)	2018-01-03-P-A

(*)해당연도에는 질병관리청 연구윤리심의위원회의 의견¹⁾에 따라 심의를 받지 않고 수행, 2018년부터 인체유래물 수집, 원시자료 제3자 제공 등을 고려 연구윤리심의 재개(문의: 건강영양조사분석과 ☎ 043-719-7494)

¹⁾국민건강영양조사는 생명윤리법 제2조 제1호 및 동법 시행규칙 제2조 제2항 제1호에 따라 국가가 직접 공공복리를 위해 수행하는 연구에 해당하여 연구윤리심의위원회 심의를 받지 않고 수행 가능

3

질병관리청은 개인정보보호법 및 통계법을 준수하여 조사 자료에서 조사 대상 가구 및 개인 추정 가능성이 높은 정보를 제거한 자료만을 제공하고 있으며, 국민건강영양조사 가속도계 원시자료는 개인정보보호를 위하여 비식별 조치로 연령을 탑코딩(top-coding) 하였습니다.

4

국민건강영양조사 원시자료 이용을 위해서는 반드시 『통계자료 이용자 준수사항 이행 서약서』 및 『보안서약서』의 서명, 제출이 필요합니다(규정 전문은 부록 참조). 해당 절차는 국민건강영양조사 누리집에서 원시자료 다운로드 시 진행됩니다.

※또한 국민건강영양조사 원시자료를 이용하여 작성된 논문, 포스터, 발표자료 등이 있는 경우에는 해당 파일을 국민건강영양조사 누리집에 등록하여 주시기 바랍니다.

※규정 전문(부록)은 개정 후 추가 공개 예정

〈원시자료 출처 표기 예시〉

구분		출처표기(Citation)
제3기 (2005)	국문	국민건강영양조사 제3기(2005), 질병관리청
	영문	The Third Korea National Health and Nutrition Examination Survey (KNHANES Ⅲ), 2005, Korea Disease Control and Prevention Agency
제8기 1차년도 (2019)	국문	국민건강영양조사 제8기 1차년도(2019), 질병관리청
	영문	The seventh Korea National Health and Nutrition Examination Survey (KNHANES Ⅶ-1), 2019, Korea Disease Control and Prevention Agency

5

본 가속도계 조사는 국민건강영양조사의 건강설문조사에 참여한 성인 대상자 중 가속도계 착용에 동의한 사람에 한하여 조사되었으며, 이는 비확률표본에 해당되어 일반화가 어려우므로 가중치 산출이 불가하여 가중치를 제공하지 않습니다. 따라서, 복합표본설계 정보(층, 집락, 가중치)를 이용하여 분석하는 국민건강영양조사 설문/검진/영양 원시자료(HN_{YR}_ALL, HN_{YR}_24RC, HN_{YR}_FFQ 등, _{YR}은 해당연도 2자리수)와 달리 가속도계 원시자료는 해당 정보를 적용할 수 없으므로 조사 참여자에 관하여만 분석 가능합니다.

6

기타 원시자료 또는 자료분석 문의는 건강영양조사과 통계팀(☎ 043-719-7467~70)으로 주시기 바랍니다.

목 차

I. 조사개요

1. 조사목적	3
2. 조사대상	3
3. 조사내용 및 방법	3
4. 참여현황	4

II. 변수설명서

1. 기본 변수	7
2. 가속도계 조사	7

III. 부록

1. SAS 프로그램	11
-------------------	----

I. 조사개요

1. 조사목적
2. 조사대상
3. 조사내용 및 방법
4. 참여현황

1. 조사목적

국민건강영양조사는 면접조사를 통해 신체활동에 대한 자료를 수집하고 있으나, 신체활동의 강도와 시간은 주관적이며 회상이 쉽지 않아 신체활동량을 정확하게 추정하는데 제한점이 있었다. 이러한 점을 보완하기 위해 제6기 2차년도(2014)부터 자기기입식 설문조사 이외에 가속도계를 도입하여 제7기 2차년도(2017)까지 신체활동에 대한 추가적인 조사를 실시하였다.

2. 조사대상

국민건강영양조사 건강설문조사 참여자(거동이 불가능한 사람 제외) 중 가속도계를 통한 신체활동량 측정에 동의한 사람을 대상으로 조사하였다. 제6기 2, 3차년도(2014-2015)와 제7기 1차년도(2016)에는 만19-64세 성인을 대상으로 하였고, 제7기 2차년도(2017)에는 노인신체활동량 측정을 위해 만65세 이상으로 선정하였다. 가속도계를 통한 신체활동량 측정에 동의한 대상자에게 실시된 조사로 대한민국 성인을 대표하는데 제한적일 수 있다.

3. 조사내용 및 방법

신체활동량 측정조사는 국민건강영양조사 중 건강설문조사를 실시한 대상자(거동이 불가능한 사람 제외)에게 해당 조사에 대해 설명하고, 동의서를 구독한 후 조사안내문 및 운동일지, 가속도계를 배부하였다. 제7기 2차년도(2017)는 자료의 누락이 많고 기입된 운동시간에 대한 보정이 불가능하여 운동일지를 제외하고 가속계만 배부하였다. 가속도계는 조사에 동의한 다음 날 부터 연속 7일 동안 배꼽을 기준으로 왼쪽(또는 오른쪽) 허리에 가속도계를 착용하되, 착용시간은 기상 직후부터 취침 직전까지 착용하도록 하였다. 가속도계는 조사에 동의한 다음 날 자정(오전 12시)부터 기록되도록 설정하였으며, 신축성이 있는 벨트에 연결하여 대상자가 맞춤 착용할 수 있도록 하였다. 또한 학교나 직장 관련 모든 활동 시 착용하되 수영, 샤워 시에는 착용하지 않도록 안내하였다.

신체활동량 측정조사에 사용된 가속도계(ActiGraph GT3X, USA)는 미국의 국민건강영양조사(NHANES) 등 국가단위조사에서 신체활동량을 측정하고 있는 도구로 신체활동으로 발생하는 가속도를 내장된 압전소자를 이용해 전기신호로 저장하는 장치이다. ActiGraph GT3X+는 3축 가속도계로써 상하(1축), 전후(2축), 좌우(3축)의 입체적 신체활동을 감지할 수 있는데 국민건강영양조사에서는 가속도계의 신체활동에 대한 강도를 상대적으로 정확하게 분류 할 수 있는 상하(1축)를 이용하였다. 따라서 대상자가 자전거타기와 같이 등가속 운동을 하였을 경우 가속도가 발생하지 않았을 수 있음을 고려해야 한다. 전기신호를 요약하는 빈도는 피험자의 신체활동 특성 및 가속도계의 저장용량 등을 고려하여 1분으로 하였다.

[분석 시 참고자료]

- 국민건강영양조사 신체활동 측정도구 결과 활용 방안 연구(질병관리청 정책연구용역보고서, 2016)
- 가속도계로 측정한 국민건강영양조사 신체활동 자료 처리방법과 활용 (한국체육측정평가학회지, 2018)
- 객관적 신체활동 측정도구 착용기간의 신뢰도추정: 2014년 국민건강영양조사 가속도계 자료분석 (한국체육측정평가학회지, 2018)

4. 참여현황

제6기 2차년도(2014)부터 제7기 2차년도(2017)까지 신체활동량 측정조사에 참여에 동의한 총 대상자수는 2,927명 이었다. 참여자수는 가속도계 분실, 미착용, 고장, 데이터 오류 등으로 참여하지 못한 대상자를 제외한 2,817명 , 참여율은 96.2% 였다. 각 연도별 대상자수 및 참여자수 연령별 분포는 표 1과 같다. 참여 대상자의 성별 비율은 2014~2016년(만 19~64세 성인 대상)에는 여자의 비율이 남자보다 약 1.5배 이상 높았고 2017년(65세이상 노인 대상)에는 남, 녀가 비슷한 비율을 보였다 (표 I).

표 I . 참여현황

(단위: 명, %)

구분	제6기						제7기					
	2차년도(2014)			3차년도(2015)			1차년도(2016)			2차년도(2017)		
	대상자수	참여자수	참여율	대상자수	참여자수	참여율	대상자수	참여자수	참여율	대상자수	참여자수	참여율
전체	1,008	977	96.9	819	791	96.6	609	575	94.4	491	474	96.5
성별												
남자	398	380	95.5	298	282	94.6	222	213	95.9	227	221	97.4
여자	610	597	97.9	521	509	97.7	387	362	93.5	264	253	95.8
연령별												
19-29	212	200	94.3	162	151	93.2	145	138	95.2	-	-	-
30-39	234	223	95.3	146	143	97.9	138	126	91.3	-	-	-
40-49	220	215	97.7	197	191	97.0	155	150	96.8	-	-	-
50-64	342	339	99.1	314	306	97.5	171	161	94.2	-	-	-
65세 이상	-	-	-	-	-	-	-	-	-	491	474	100.0

II. 변수설명서

1. 기본 변수
2. 가속도계 조사

1. 기본 변수

변수유형	변수명	변수설명	내용
C(12)	mod_d	최종 DB 수정일	
C(10)	ID	개인 아이디	A 0 0 1 0 0 0 1 0 1 조사구번호 거처·주거번호 가구원 번호
N(4)	year	조사연도	
N(1)	sex	성별	1. 남자 2. 여자
N(3)	age	만나이 ¹⁾	세 1~79. 1~79세 80. 80세 이상

1) 제7기 2차년도(2017)는 비식별조치방법 중 탑코딩(top-coding) 적용

2. 가속도계 조사

변수유형	변수명	변수설명	내용
N(1)	day	착용일	1. 착용 1일차 2. 착용 2일차 3. 착용 3일차 4. 착용 4일차 5. 착용 5일차 6. 착용 6일차 7. 착용 7일차
N(1)	PAXDAY	착용 요일	1. 일요일 2. 월요일 3. 화요일 4. 수요일 5. 목요일 6. 금요일 7. 토요일
N(5)	PAXN	개인 아이디별 측정 순서	1)
N(2)	PAXHOUR	측정 시간(시)	시 0~23. 0~23시
N(2)	PAXMINUT	측정 시간(분)	분 0~59. 0~59분
N(5)	PAXINTEN	측정 신체활동 강도 ²⁾	
N(3)	PAXSTEP	측정 걸음 수 ³⁾	

1) 가속도계에 순차적으로 기록되는 번호로써 1은 측정 첫날 1분을 의미하며 측정 7일차 마지막 1분은 10080으로 기록

2) 요약된 전기 신호의 크기이고, 크기에 따라 해당 시간 동안 대상자의 작업 활동, 저항도, 중등도, 고강도 신체활동을 구분 할 수 있음

3) 1분 동안 걸음수

Ⅲ. 부 록

1. SAS 프로그램

1. SAS 프로그램

■ NHANES(미국, National Health and Nutrition Examination Survey) 분석 프로그램¹⁾

■ 미착용 시간과 착용 시간을 구분하는 매크로 %nw(nwperiod=) 생성

Macro %nw defines the duration of non-wear periods as well as wear periods within a day based on user defined minimum length of non-wear period.

A non-wear period starts at a minute with the intensity count of zero. Minutes with intensity count=0 or up to 2 consecutive minutes with intensity counts between 1 and 99 are considered to be valid non-wear minutes. A non-wear period is established when the specified length of consecutive non-wear minutes is reached. The non-wear period stops when any of the following conditions is met:

- one minute with intensity count >99
- one minute with a missing intensity count
- 3 consecutive minutes with intensity counts between 1 and 99
- the last minute of the day

Macro call %nw(nwperiod=);

nwperiod: minimum length for the non-wear period, must be >1 minute.

```
%macro nw(nwperiod=);
data nw_all; set monitors;
  by id day paxn;

  if first.day then nw_num=0;          /*non-wear period number*/

  if first.day or reset or stopped then do;
    strt_nw=0;          /*starting minute for the non-wear period*/
    end_nw=0;          /*ending minute for the non-wear period*/
    start=0;          /*indicator for starting to count the non-wear period*/
    dur_nw=0;          /*duration for the non-wear period*/
    reset=0;          /*indicator for resetting and starting over*/
    stopped=0;          /*indicator for stopping the non-wear period*/
    cnt_non_zero=0;          /*counter for the number of minutes with intensity between 1 and 99*/
  end;
  retain nw_num strt_nw end_nw stopped reset start cnt_non_zero dur_nw;

  /*The non-wear period starts with a zero count*/
  if PAXINTEN=0 and start=0 then do;
    strt_nw=paxn;          /*assign the starting minute of non-wear*/
    start=1;
  end;

  /*accumulate the number of the non-wear minutes*/
  if start and PAXINTEN=0 then
    end_nw=paxn;          /*keep track of the ending minute for the non-wear period*/

  /*keep track of the number of minutes with intensity between 1-99*/
  if 0<PAXINTEN<=99 then
    cnt_non_zero=cnt_non_zero+1;

  /*before reaching the 3 consecutive minutes of 1-99 intensity, if encounter one minute with zero intensity,
  reset the counter*/
  if PAXINTEN=0 then cnt_non_zero=0;
/*duration of non-wear period*/
```

1) 미국의 NHANES(National Health and Nutrition Examination Survey)에서 가속도계 원시자료와 함께 공개한 분석 프로그램으로 국민건강영양조사 가속도계 원시자료 분석시 참고자료로 활용


```

dur_nw=end_nw-strt_nw+1;

/*A non-wear period ends with 3 consecutive minutes of 1-99 intensity, one missing count, or one minute
with >99 intensity*/
if (cnt_non_zero=3 or PAXINTEN=. or PAXINTEN>99 ) then do;
  if dur_nw<&nwperiod then reset=1;          /*reset if less than &nwperiod minutes of non-wear*/
  else stopped=1;
end;

/*last minute of the day*/
if last.day and dur_nw>=&nwperiod then stopped=1;

/*output one record for each non-wear period*/
if stopped=1 then do;
  nw_num=nw_num+1;
  keep id day nw_num strt_nw end_nw dur_nw;
  output;
end;
run;

```

■ id별 착용일별 미착용 시간(분) 총합 계산

summarize the non-wear periods to one record per day

```

proc summary data=nw_all;
  by id day;
  var dur_nw;
  output out=sum_nw
         sum=tot_dur_nw;
run;

```

■ id별 착용일별 분석 가능한 착용 시간(분) 총 합 계산

summarize the total number of valid minutes for everyone in the analysis

```

proc summary data=monitors;
  by id day age;
  var PAXINTEN;
  output out=sum_all
         n=tot_min;
run;

```

■ id별 착용일별 미착용 시작 시점(strt_nw)부터 미착용 종료 시점(end_nw)까지 1씩 증가하는 변수를 포함하는 DB 생성

create a dataset with one record per minute, for non-wear periods only

```

data nw_minutes(keep=id day paxn);
  set nw_all;
  by id day nw_num;
  do i=strt_nw to end_nw by 1;
    paxn=i;
    output;
  end;
run;

```

■ id별 가속도계를 착용한 시간만 포함하는 DB 생성

create a dataset with one record per minute, for wear periods only

```
data wear_minute(keep=id day paxn PAXINTEN);
  merge monitors(in=in_all) nw_minutes(in=in_nw);
  by id day paxn;
  if in_all and not in_nw;
run;
```

/*summarize information from wear minutes*/

```
proc summary data=wear_minute;
  by id day;
  var PAXINTEN;
  output out=sum_wear
    sum=tot_cnt_wr
    n=tot_min_wr;
run;
```

■ id별 착용일별 요약 변수를 모두 포함하는 DB 생성

creates dataset with one record per day for everyone with monitor data that includes summary variables from the non-wear algorithm

```
data nw&nwperiod;
  merge sum_all(in=in_all) sum_nw(in=in_nw) sum_wear;
  by id day;
  if in_all;

  if tot_dur_nw=. then tot_dur_nw=0;
  if tot_min_wr=. then tot_min_wr=0;
  if tot_cnt_wr=. then tot_cnt_wr=0;

  wear_hr=tot_min_wr/60;
  tot_dur_nw=tot_dur_nw/60;

  if epoch=120 then do;
    tot_dur_nw=(tot_dur_nw*2); wear_hr=(wear_hr*2); tot_min=(tot_min*2); tot_cnt_wr=(tot_cnt_wr*2);
    tot_min_wr=(tot_min_wr*2);
  end;

  if epoch=10 then do;
    tot_dur_nw=(tot_dur_nw*1/6); wear_hr=(wear_hr*1/6); tot_min=(tot_min*1/6); tot_cnt_wr=(tot_cnt_wr*1/6);
    tot_min_wr=(tot_min_wr*1/6);
  end;

  label
    tot_dur_nw='Total duration(hr) of non-wear periods in a day'
    wear_hr='Total number of wear hours for the day'
    tot_min='Total number of valid minutes within a day'
    tot_cnt_wr='Total intensity counts from all wear minutes in a day'
    tot_min_wr='Total number of wear minutes in a day';
  keep id day age tot_min tot_min_wr wear_hr tot_cnt_wr tot_dur_nw;
run;

%mend nw;

%nw(nwperiod=60);          /*this is where the duration criterion for a non-wear period is set*/
```

■ id별 시간(분)에 따른 운동 여부 DB 생성

Makes a shell data set then same size as monitors that generates a data flag for every minute as a wear minute or a nonwear minute. This flag is merged into the monitors dataset and will be used below to classify intensity. The shell data set is then deleted.

```
data wear_shell;
  retain last_stop 0;
  set nw_all(drop=day nw_num dur_nw);
  by id;
  if first.id then do;
    last_stop=0;
    end;
  do paxn=last_stop+1 to strt_nw-1;
    __wear__ = 1;
    output;
    end;
  do paxn=strt_nw to end_nw;
    __wear__ = 0;
    output;
    end;
  last_stop = end_nw;
  if last.id then do;
    if (end_nw < 10080)
      then do paxn=end_nw+1 to 10080;
        __wear__=1;
        output;
        end;
    end;
  keep id paxn __wear__;
run;

data monitors;
  merge monitors wear_shell;
  by id paxn;
run;

proc datasets nolist;
  delete wear_shell;
run;
```

■ 대상자별 측정값을 비교하기 위한 활동 강도별 기준값 정의

Code below classifies intensity of each minute of monitor data by comparing paxinten to user-definable intensity thresholds. In some cases, intensity criteria are age-specific.

NOTE: to change the intensity thresholds, modify the statements below for these variables
 sedthresh (sedentary threshold)
 lighththresh (light threshold)
 lifethresh (lifestyle moderate threshold)
 modthresh (moderate threshold) and
 vigthresh (vigorous threshold).

```
data monitors;
  set monitors;
  * sedentary threshold *;           /*앉아서 하는 또는 몸을 많이 움직이지 않는 활동 기준값 정의*/
  sedthresh=0;

  * light threshold *;              /*가벼운 활동 기준값 정의*/
  lighththresh = 100;

  * lifestyle moderate threshold is only defined for adults *;      /*성인 기준 일상생활에서의 활동 기준값 정의*/
  lifethresh = 760;
```

```

* moderate threshold *;          /*보통 강도의 활동 기준값 정의*/
modthresh=2020;

* vigorous threshold *;          /*격렬한 활동 기준값 정의*/
vigthresh=5999;

* sedentary activity *;          /*앉아서 하는 또는 몸을 많이 움직이지 않는 활동 여부*/
if (__wear__ = 1 and sedthresh <= PAXINTEN < lightthresh) then _s=1;
else if PAXINTEN ne . then _s=0;

* light activity *;              /*가벼운 활동 여부*/
if (lightthresh <= PAXINTEN < lifethresh) then _l=1;
else if PAXINTEN ne . then _l=0;

* lifestyle moderate activity - for adults only *;          /*성인 기준 일상생활에서의 활동 여부*/
if age >= 18 then do;
  if (lifethresh <= PAXINTEN < modthresh) then _ls=1;
  else if PAXINTEN ne . then _ls=0;
end;

* moderate or vigorous activity *;          /*보통 강도 또는 격렬한 활동 여부*/
if PAXINTEN >= modthresh then _mv=1;
else if PAXINTEN ne . then _mv=0;

* moderate activity *;          /*보통 강도 활동 여부*/
if modthresh <= PAXINTEN < vigthresh then _m=1;
else if PAXINTEN ne . then _m=0;

* vigorous activity *;          /*격렬한 활동 여부*/
if PAXINTEN >= vigthresh then _v=1;
else if PAXINTEN ne . then _v=0;

* non-wear *;                  /*미착용 여부*/
if __wear__ = 0 then _nw=1;
else _nw=0;

run;

proc sort data=monitors;
  by id day paxn;
run;

```

■ 측정된 운동 강도가 연속적으로 특정값 이상인 경우를 만족하는 구간 계산

Activity bouts are defined by specified number of consecutive minutes with intensity count $\geq$ the relevant intensity threshold.
Two separate bout macros are used below, then first flags bouts of 1 minute or more to allow a sum of all minutes above threshold. The second looks for bouts of 8-10 minutes or more.

An activity bout starts a minute with an intensity count greater than or equal to the threshold. Minutes with intensity count greater than or equal to the threshold are considered to be valid minutes for the activity bout. A bout is established when then specified length of consecutive valid minutes are reached. The activity bout stops when any of then following conditions is met:

- one minute with intensity $<$ threshold
- one minute with a missing intensity count
- the last minute of the day

Macro %bouts defines duration of activity bouts based on user defined minimum bout length.

Macro call %bouts(bout_flg=boutperiod=);

bout_flg: variable name for activity bout intensity: _s(sedentary), _l(light), _ls(lifestyle moderate), mv(moderate or vigorous), _m(moderate), _v(vigorous), and _nw(non-wear)
boutperiod: minimum bout length(1 minute, 2 minutes, 3 minutes, etc)

```
%macro bouts(bout_flg=boutperiod=);
data out&bout_flg&boutperiod;
  set monitors;
  by id day paxn;
  if first.day then mv_num=0;          * number of activity bouts *;

  if first.day or reset or stopped then do;
    strt_mv=0;          * starting minute for the activity bout *;
    end_mv=0;          * ending minute for the activity bout *;
    start=0;          * indicator for starting the activity bout *;
    reset=0;          * indicator for resetting and starting over *;
    mv_cnt=0;          * number of minutes for the activity bout *;
    stopped=0;          * indicator for stopping the activity bout *;
  end;
  retain mv_num strt_mv end_mv mv_cnt stopped reset start;

  * start the bout when a count with intensity >= the threshold is encountered *;
  if &bout_flg=1 and start=0 then do;    /*default start=0*/
    strt_mv=paxn;          * assign the starting minute of the bout *;
    start=1;
  end;

  * accumulate minutes with intensity counts >= the threshold *;
  if start=1 and &bout_flg=1 then do;
    mv_cnt=mv_cnt+1;
    end_mv=paxn;          * keep track of the ending minute for the bout *;
  end;

  * stop when encounter a minute with intensity < threshold or missing *;
  if &bout_flg in (0,.) then do;
    if mv_cnt<&boutperiod then reset=1;    * reset if less than the bout length *;
    else stopped=1;
  end;

  * last minute of the day *;
  if last.day and mv_cnt>=&boutperiod then stopped=1;

  * output one record for each activity bout *;
  if stopped=1 then do;
    dur_mv=end_mv-strt_mv+1;
    mv_num=mv_num+1;
    output;
  label
    strt_mv='Starting minute for the activity bout'
    end_mv='Ending minute for the activity bout'
    dur_mv='Duration(minutes) of activity bout'
    mv_num='Number of activity bout';
  end;
  keep id day mv_num strt_mv end_mv dur_mv;
run;

proc sort data=out&bout_flg&boutperiod;
  by id day mv_num;
```

```
run;
```

■ 착용일별 운동 시간(분) 총합 계산

calculate total duration of activity bouts for each intensity for each day.

```
proc summary data=out&bout_flg&boutperiod;  
  by id day;  
  var dur_mv;  
  output out=sum_mv  
         sum=tot_dur_mv;  
run;
```

■ id별 착용일별 분석 결과 출력

output one record per day for each person

```
data out&bout_flg&boutperiod._sum;  
  merge sum_all(in=in_all) sum_mv;  
  by id day;  
  if in_all;  
  if tot_dur_mv=. then tot_dur_mv=0;  
  label  
  %if &bout_flg=_s %then %do;  
    tot_dur_mv="Total duration(minutes) of sedentary activity bouts (minimum &boutperiod minute bouts) in a day"  
  %end;  
  %if &bout_flg=_l %then %do;  
    tot_dur_mv="Total duration(minutes) of light activity bouts (minimum &boutperiod minute bouts) in a day"  
  %end;  
  %if &bout_flg=_ls %then %do;  
    tot_dur_mv="Total duration(minutes) of lifestyle activity bouts (minimum &boutperiod minute bouts) in a day"  
  %end;  
  %if &bout_flg=_mv %then %do;  
    tot_dur_mv="Total duration(minutes) of moderate or vigorous activity bouts (minimum &boutperiod minute bouts) in a day"  
  %end;  
  %if &bout_flg=_m %then %do;  
    tot_dur_mv="Total duration(minutes) of moderate activity bouts (minimum &boutperiod minute bouts) in a day"  
  %end;  
  %if &bout_flg=_v %then %do;  
    tot_dur_mv="Total duration(minutes) of vigorous activity bouts (minimum &boutperiod minute bouts) in a day"  
  %end;  
  ;  
  
  if epoch=120 then tot_dur_mv=(tot_dur_mv*2);      *120s;  
  if epoch=10 then tot_dur_mv=(tot_dur_mv*1/6);      *10s;  
  
  keep id day tot_dur_mv;  
  rename  
  tot_dur_mv=tot_dur&bout_flg&boutperiod;  
run;  
%mend bouts;
```

■ 매크로 명령문(1분으로 설정되어 있으나 연구자의 분석 의도에 따라 변경하여 분석 가능)

the macro statements below set bout length (currently set to 1 min, which accumulates every minute within an intensity category)

```
%bouts(bout_flg=_s, boutperiod=1);
%bouts(bout_flg=_l, boutperiod=1);
%bouts(bout_flg=_ls, boutperiod=1);
%bouts(bout_flg=_mv, boutperiod=1);
%bouts(bout_flg=_m, boutperiod=1);
%bouts(bout_flg=_v, boutperiod=1);
%bouts(bout_flg=_nw, boutperiod=1);
```

■ 측정된 운동 강도가 특정값 이상인 경우가 10분 중 8분 이상 지속되는 구간 계산

Macro %bouts_8of10 defines activity bouts for 8 out 10 minutes with intensity count $\geq$ the specified threshold.

An activity bout starts with a count that is greater than or equal to the threshold. A bout is established when 8 minutes out of a 10 minute window have intensity counts greater than or equal to the threshold. The bout stops when any of the following conditions in met:

- 3 consecutive minutes with intensity $<$ threshold
- one minute with a missing intensity count
- last minute of the day

Macro call %bouts_8of10(bout_flg=);

bout_flg: variable name for activity bout intensity, _m(moderate), v(vigorous), _mv(moderate or vigorous) other intensities can be specified for inclusion.

```
%macro bouts_8of10(bout_flg=);
data out&bout_flg;    set monitors;
by id day paxn;

* set up a 10 minute window *;
array win_paxn(*) win_paxn1-win_paxn10;    * minute *;
array win_int(*) win_int1-win_int10;    * intensity *;
array win_flg(*) win_flg1-win_flg10;    * bout flag *;

if first.day then
    mv_num=0;    * number of activity bouts *;

if first.day or stopped or reset then do;
    strt_mv=0;    * starting minute for the bout *;
    end_mv=0;    * ending minute for the bout *;
    found=0;    * set to 1 if a bout has been established *;
    reset=0;    * reset the counts and start over *;
    stopped=0;    * indicator for stopping the bout *;
    start=0;    * start set to 1 if one above the threshold count is encountered *;
    mv_cnt=0;    * number of minutes with counts  $\geq$  the threshold *;
    sum10=.;    * the total intensity counts from the 10 minute window *;
    cnt_below=0;    * counter for number of minutes with intensity below the threshold *;
    do i=1 to 10;    * initialize the 10 minute window *;
        win_paxn(i)=0;
        win_int(i)=0;
        win_flg(i)=0;
    end;
end;
retain mv_num reset strt_mv end_mv start mv_cnt found stopped sum10 cnt_below;
retain win_paxn1-win_paxn10;
retain win_int1-win_int10;
retain win_flg1-win_flg10;
* if the intensity count is  $\geq$  the threshold, start the bout *;
```

```

if &bout_flg=1 and start=0 then
    start=1;

* accumulate the counts *;
if start=1 then mv_cnt=mv_cnt+1;

* set up a moving window of 10 minutes *;
if 1<=mv_cnt<=10 and not found then do;
    win_paxn(mv_cnt)=paxn;
    win_int(mv_cnt)=paxinten;
    win_flg(mv_cnt)=&bout_flg;
    if paxinten = . then reset=1;
* if encounter a missing count before reaching the 10 minute count, reset and start again *;
end;

* when reach 10 minutes, count the total number of intensity counts that are >= threshold *;
if mv_cnt=10 and not reset then sum10=sum(of win_flg1-win_flg10);

* if 8 out of 10 minutes with intensity counts >= the threshold, a bout is established *;
if sum10>=8 then found=1;

* if less than 8 minutes with intensity counts>= the threshold, continue to search *;
* move the 10-minute window down, one minute at a time *;
else if 0<sum10<8 and mv_cnt>10 then do;
    if paxinten=. then reset=1;          * if the 10th minute has a missing count, reset and start again *;
    else do;
        do i=1 to 9;
            win_paxn(i)=win_paxn(i+1);
            win_int(i)=win_int(i+1);
            win_flg(i)=win_flg(i+1);
        end;
        * read in minute 10 *;
        win_paxn(10)=paxn;
        win_int(10)=paxinten;
        win_flg(10)=&bout_flg;
        sum10=sum(of win_flg1-win_flg10);
    end;
end;
if sum10 in (0) then reset=1;          * skip the windows with no valid minutes *;

* after the bout is established *;
if found then do;

    * assign the starting minute for the activity bout *;
    if strt_mv= 0 then do;
        do i=1 to 10;
            if win_flg(i)=1 then do;          * find the first minute with intensity count>=the threshold *;
                strt_mv=win_paxn(i);
                i=11;
            end;
        end;
    end;

    * assign the ending minute for the activity bout *;
    if end_mv=0 then do;
        * the last 2 minutes in the 10 minute window are below the threshold *;
        if win_flg(9)=0 and win_flg(10)=0 then do;
            end_mv= win_paxn(8);
            cnt_below=2;
        end;

        * the last minute in the 10 minute window is below the threshold *;
    end;
end;

```



```

        else if win_flg(10)=0 then do;
            end_mv=win_paxn(9);
            cnt_below=1;
        end;
    else
        end_mv=win_paxn(10);
    end;

    if paxn>win_paxn(10) then do;
        if &bout_flg=1 then do;
            cnt_below=0;
            end_mv=paxn;
        end;
        if &bout_flg=0 then
            cnt_below=cnt_below+1;
        * keep track of the number of minutes with intensity counts below the threshold *;
    end;
    * bout terminates if 3 consecutive minutes below the threshold are encountered, or a missing count,
    or the last minute of the day *;
    if cnt_below=3 or last.day or &bout_flg=. then stopped=1;
end;

* output one record for each activity bout *;
if stopped=1 then do;
    dur_mv=end_mv-strt_mv+1;
    mv_num=mv_num+1;
    keep id day mv_num strt_mv end_mv dur_mv;
    output;
end;
run;
proc sort data=out&bout_flg;
    by id day mv_num;
run;

```

■ 착용일별 운동 시간(분) 총합 계산

calculate total duration of activity bouts for each day.

```

proc summary data=out&bout_flg;
    by id day;
    var dur_mv;
    output out=sum_mv
           sum=tot_dur_mv;
run;

```

■ 개인 id별 측정일별 분석 결과 출력

output one record per day for each person in the analysis.

```

data out&bout_flg_sum;
    merge sum_all(in=in_all) sum_mv;
    by id day;
    if in_all;
    if tot_dur_mv=. then tot_dur_mv=0;
    label
    %if &bout_flg=_mv %then %do;
        tot_dur_mv="Total duration(min) of moderate or vigorous activity bouts (8 out of 10 minutes) in a day"
    %end;
    %if &bout_flg=_m %then %do;
        tot_dur_mv="Total duration(min) of moderate activity bouts (8 out of 10 minutes) in a day"
    %end;

```

```

%if &bout_flg=_v %then %do;
  tot_dur_mv="Total duration(min) of vigorous activity bouts (8 out of 10 minutes) in a day"
%end;
;

  if epoch=120 then tot_dur_mv=(tot_dur_mv*2); *120s;
  if epoch=10 then tot_dur_mv=(tot_dur_mv*1/6); *10s;

keep id day tot_dur_mv;

rename
  tot_dur_mv=tot_dur&bout_flg;
run;
%mend bouts_8of10;

%bouts_8of10(bout_flg=_mv);
%bouts_8of10(bout_flg=_v);
%bouts_8of10(bout_flg=_m);

```